
Overview
• Proc SGSCATTER, SGPLOT, SGPANEL and SGRENDER are part of 

the SAS ODS graphics procedures also called Statistical Graphics 
procedures

• SGSCATTER – creates scatter plots
• SGPLOT – creates single cell plots with a variety of plot and chart 

types and overlays
• SGPANEL – creates classification panels for one or more panel 

variables
• SGRENDER – produces graphs from graph templates that are 

written in Graph Template Language (GTL)
• For a full list of SAS ODS graphics procedures please see the 

reference section at the end of the presentation

Demo 1: Can we predict iris sepal length based on iris sepal width?
• Sepals cover buds and are found immediately below the flower 

after blooming
• SASHELP.IRIS – Ronald Fisher's and Edgar Anderson's sample of 

50 of each of 3 iris species to create a linear discriminant analysis 
model to distinguish species from each other

Let's look at the data for the iris Setosa species
• Is sepal width normally distributed?
%let species=Setosa;

proc sgplot

data=sashelp.iris(where=(species="&species."));

title "Sepal Width Frequency Distribution for 

&species.";

histogram sepalwidth / scale=count;

density sepalwidth / scale=count;

run;

title;

Distribution looks normal

Let's look at the data for the iris Setosa species
• Does sepal width have a linear relationship with sepal length?
• Do we see homoskedasticity? i.e. do variances look consistent 

across all pairs of sepal width and sepal length

%let species=Setosa;

proc sgscatter

data=sashelp.iris(where=(species="&species."));

plot sepallength*sepalwidth / reg;

title "Sgscatter scatter plot of Sepal Width 

(x) against Sepal Length (y) for &species.";

run;

• Regression line and data points suggests linear relationship 
• Homoskedastic - variance is consistent.  No cone at either end

If assumptions are met, we can perform simple linear regression to 
predict sepal length from sepal width

%let species=Setosa;

proc reg

data=sashelp.iris(where=(species="&species.")) 

outest=est1;

eq1: model sepallength=sepalwidth;

run;

Analysis of Variance

Source DF
Sum of

Squares
Mean

Square F Value Pr > F

Model 1 335.68848 335.68848 58.99 <.0001

Error 48 273.13152 5.69024

Corrected Total 49 608.82000

Root MSE 2.38542 R-Square 0.5514

Dependent Mean 50.06000 Adj R-Sq 0.5420

Coeff Var 4.76513

Parameter Estimates

Variable Label DF
Parameter

Estimate
Standard

Error t Value Pr > |t|

Intercept Intercept 1 26.39001 3.10014 8.51 <.0001

SepalWidth Sepal Width (mm) 1 0.69049 0.08990 7.68 <.0001

Create scatterplot using sgplot instead of sgscatter
• Add confidence limits for the mean CLM and individual predicted 

values CLI

%let species=Setosa;

proc sgplot

data=sashelp.iris(where=(species="&species."));

scatter x=sepalwidth y=sepallength;

title "Sgplot scatter plot of Sepal Width (x) 

against Sepal Length (y)";

reg x=sepalwidth y=sepallength /clm cli;

run;

Modify sgplot code to create a panel plot for all species
• Change sgplot to sgpanel
• Add panelby statement for species

proc sgpanel data=sashelp.iris;

panelby species;

scatter x=sepalwidth y=sepallength;

title "Scatter plot of Sepal Width (x) 

against Sepal Length (y)";

reg x=sepalwidth y=sepallength /clm cli;

run;

Demo 2:  Compare sales data by quarter between 1999 and 2002
• Use SASHELP.ORSALES
• Aggregate and format input data for line chart with 2 y axes

proc sql;

create table orsales_qtr as

select year

,substr(quarter,5,2) as qtr format $2.

,sum(profit) as profit format dollar13.

,sum(quantity) as quantity format comma15.

from sashelp.orsales

group by year,calculated qtr

order by year,calculated qtr;

quit;

Use SGPLOT to create a line chart with 2 y axes

proc sgplot data=orsales_qtr;

title;

title1 color=black "Orion Sales 1999 - 2002";

styleattrs datacontrastcolors=(purple green orange 

blue);

xaxis type=discrete label='Quarter';

yaxis label='Units Sold - Solid Line' grid minor;

y2axis label='Profit $ - Dashed Line' minor;

series x=qtr y=quantity / group=year 

lineattrs=(pattern=solid);

series x=qtr y=profit / group=year 

lineattrs=(pattern=longdash) y2axis;

INSET 'Units Sold and Profit by Quarter'/ POSITION

= BOTTOM BORDER TEXTATTRS=(Size=11 Weight=Bold);

run;



Add SGSCATTER, SGPLOT, SGPANEL and SGRENDER Procedures to Your SAS Toolbelt

Use SGPLOT to create line chart with 2 y axes
• Use one color for each of four years, datacontrastcolors
• Use solid line for quantity, lineattrs=(pattern=solid)          
• Use dashed line for profit, lineattrs=(pattern=longdash)        
• Group lines by year, group=year                                 
• Use yaxis and y2axis to reference left and right hand vertical axes  
• Inset description at bottom of plot, inset 'text' / position=bottom 

Orion Sales 1999 - 2002

Demo 3:  Plotting in 3 dimensions with SGRENDER 
Example from  SAS® 9.4 ODS Graphics: Procedures Guide, Sixth 
Edition
/* Create stat graph template */

proc template;

define statgraph surface;

begingraph;

layout overlay3d;

surfaceplotparm x=height y=weight 

z=density;

endlayout;

endgraph;

end;

run;

Plotting in 3 dimensions using SGRENDER 
Generate graphics output from the template
• SASHELP.GRIDDED contains information on height, weight and 

density

title;

title1 'Height, weight and density plot based 

on a custom statgraph template';

proc sgrender data=sashelp.gridded

template=surface;

run;

title;

title1

Height, weight and density plot based on a custom statgraph template

QUESTIONS


